Importing Verilog to Cadence Schematic

To create a Cadence schematic from structural verilog, you must write all of your verilog code calling modules in your cell library. The following is an example of the library needed to implement a 32 bit ripple carry adder.



When these modules are imported into cadence, they will not produce, schematic views. The schematics for behavioral models must be created manulally. The following modules form the 32 bit adder and are all defined in terms of the the cell library. Be careful when you name your nets because these nets will become the spice nets -- and spice is case insencitive.




```
 ****************************************
 * these modules are structural
 * and will create schematic views
 ****************************************
 ****************************************/

module xor2(out, a, b);
   output out;
   input  a, b;
   wire   a_nor_b;
   nor2 nor2(a_nor_b, a, b);
   aoi12 aoi12(out, a_nor_b, a, b);
endmodule // xor2

module xor3(out, a, b, c);
   output out;
   input  a, b, c;
   wire   a_xor_b;
   xor2 xor_0(a_xor_b, a, b);
   xor2 xor_1(out, a_xor_b, c);
endmodule // xor3

module ao222(out, a, b, c);
   output out;
   input  a, b, c;
   wire   a_nand_b, a_nand_c, b_nand_c;
   nand2 nand_0(a_nand_b, a, b);
   nand2 nand_1(a_nand_c, a, c);
   nand2 nand_2(b_nand_c, b, c);
   nand3 nand_3(out, a_nand_b, a_nand_c, b_nand_c);
endmodule // ao222

module full_adder(sum, cout, a, b, cin);
   output sum, cout;
   input  a, b ,cin;

   xor3 xor_sum(sum, a, b, cin);
   ao222 ao_carry(cout, a, b, cin);
endmodule // full_adder

module adder_8b(sum, cout, a, b, cin);
   output [7:0] sum;
   output       cout;
   input [7:0]  a, b;
   input        cin;
   wire [7:1]   carry;

   full_adder adder0(sum[0], carry[1], a[0], b[0],      cin);
   full_adder adder1(sum[1], carry[2], a[1], b[1], carry[1]);
   full_adder adder2(sum[2], carry[3], a[2], b[2], carry[2]);
   full_adder adder3(sum[3], carry[4], a[3], b[3], carry[3]);
   full_adder adder4(sum[4], carry[5], a[4], b[4], carry[4]);
   full_adder adder5(sum[5], carry[6], a[5], b[5], carry[5]);
   full_adder adder6(sum[6], carry[7], a[6], b[6], carry[6]);
   full_adder adder7(sum[7],     cout, a[7], b[7], carry[7]);

endmodule // adder_8b

module adder_32b(sum, cout, a, b, cin);
   output [31:0] sum;
   output        cout;
   input [31:0]  a, b;
   input         cin;
   wire [3:1]    carry;

   adder_8b adder0(sum[07:00], carry[1], a[07:00], b[07:00],      cin);
   adder_8b adder1(sum[15:08], carry[2], a[15:08], b[15:08], carry[1]);
   adder_8b adder2(sum[23:16], carry[3], a[23:16], b[23:16], carry[2]);
   adder_8b adder3(sum[31:24],     cout, a[31:24], b[31:24], carry[3]);
endmodule // adder_32b
```

After you have tested and debugged your verilog, you may import it into a Cadence schematic. Open Cadence with

> layoutPlus&

In tha main window, choose File -> Import -> Verilog



This will open the Verilog In window.



Input the correct Target Libray Name. Clear the Refrence Library Field. Browse the directory tree to find your verilog source file. You can put *.v into the File Filter Name to only show verilog source files. Select the desired source file (ripple.v in this example) and press the Add button for Verilog files to import. You may want to select the Overwrite Existing Views button if you want to change any of the schematics. Select Apply.



You will see a popup window asking you if you want to see the log file. Select Yes. If you did not see this popup, check the main window for errors.



In the Log File, you can see that that the symbol views were created for all of the modules, Schematic vies were created for the structural modules, and functional views were created for the behavioral views (the functional view just contains the text source.) If you have already created schematic views for the behavioral modules and the pin names do not match, you will have to changes the net names.

To see the newly improrted cells, you may have to click refresh in the Library Manager.



Open the top level schematic to check the design.



Choose Check -> Hierarchy.



Check every schematic and press OK. Look in the main window for errors. Netlist the design as usual except that it is a good idea to select Re-netlist Entire Design in the simulation options.



When you netlist the design, check Re-netlist Entire Design just to be sure that you are using the current versions of all of your cells. In the example design, the modules nand2, nand3, aoi12, nor2 must all be implemented at the transistor level and will not netlist until these schematics are defined manually. If you have not yet defined the low-level cells, you will get the folling error while netlisting.



If you ever have to modify the verilog code and re-import the design, you should delete all of the cells that you want to update using the library manager. Then always check the entire hierarchy for errors.

Tips

Always remember that verilog is case sensitive but spice is not.

Dont use reserved keywords in verilog to name modules or wires -- and, xor, ...

Vectors in verilog -- a[31:0] -- will turn into busses in schematic -- a<31:0> -- and will netlist into indivitual nets in spice -- a31, a31, ... a0. If your verilog source has a net called a0, you will have problems.

Debug your verilog before trying to import the design to schematic.

If your verilog modules contains any behavioral code -- includeing assign statements -- it will not produce a schematic.

Sometimes the schematic will produduce routing errors if you module is very complex so keep modules small and simple.